



Weld your way.

Карола -

Требования

Требования к коду
программы Carola

Содержание

1 Введение	3
2 Общая эффективность оборудования (ОЕЕ)	3
2.1 Определите фактическое количество и хорошее количество	3
2.1.1 Магистерская программа	3
2.1.2 Переменные	6
2.1.3 Простая компонентная программа	7
2.1.4 Комплексная компонентная программа	8
3 Примечания	10
4 Переменные	10
5 Условия	11

1 Введение

Для некоторых функций C-Gate необходимо, чтобы информация передавалась на C-Gate из программы Carola. Необходимые блоки и инструкции для соответствующей функциональности см. в соответствующих главах.

Если вы используете структуру данных `ReadVar` или команду `OPCUAVAR` в другом контексте, убедитесь, что используемые области не пересекаются. Следует также учитывать, что индекс, используемый в программе Carola, не является тем же самым индексом в веб-интерфейсе. Значения передаются в C-Gate через структуру данных `ReadVar`. В контроллере первый индекс равен единице, а последний 100. В C-Gate первый индекс равен 0, а последний 99. Значение, хранящееся в контроллере под индексом 50, может быть получено в C-Gate под индексом 49.

Обратите внимание, что примеры кода Carola, приведенные ниже, могут быть частично или полностью уже сохранены в магистерской программе. Пожалуйста, сравните разделы и при необходимости адаптируйте их. Дублирование функций и переменных распознается интерпретатором Carola как ошибка.

Примечание: Изменение переменных `QIROX WriteVar` может повлиять на последовательность программ Carola.

Примечание: Для копирования программного кода Carola используйте Adobe Acrobat Reader для открытия этого файла.

Примечание: Пожалуйста, используйте строки длиной не более 32 символов. В противном случае значения будут обработаны только частично или не будут обработаны вовсе.

2 Общая эффективность оборудования (ОЕЕ)

2.1 Определите фактическое количество и хорошее количество

Для определения фактического и допустимого количества в мастер- и компонентной программе должны быть сохранены следующие разделы программы соответственно. Важно, что в следующих процедурах используется исключительно диапазон индексов от 94 до 100 и не может быть использован для других целей. В будущем этот диапазон может расшириться.

2.1.1 Магистерская программа

Первые строки необходимы для того, чтобы переменные можно было импортировать и записывать из других программ. Они располагаются непосредственно после инструкции `RESTART`.

```
RESTART
! ОБЪЯВЛЕНИЕ ПЕРЕМЕННОЙ
VAR SEL, RESULT
STRING VAR COMPSTAT, COMPQUAL
PUBLIC STRING VAR COMPTYPE, COMPVARI, COMPSERI, COMPSTEP, COMPSEAM, SEAMQUAL
```

Этот блок определяет процедуру, которая посылает на C-Gate информацию о запуске компонента. Перед этим в программе компонента должны быть установлены переменные, используемые ниже.

```

PUBLIC PROC COMPSTART
COMPSTAT := ''
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
WAITM(500)

STRHND(2, COMPTYPE, COMPTYPE, RESULT)
IF RESULT=0 THEN COMPTYPE:='NODATA'
STRHND (0,COMPTYPE, 'NODATA',RESULT)
IF RESULT=1 THEN BEGIN
LOPCUA1:
WRITE ('ПЕРЕМЕННАЯ >COMPTYPE< ЕЩЕ НЕ ОПИСАНА!')
WAITM(500)
PAUSE
JUMP LOPCUA1
END

COMPSTAT := 'START'
COMPQUAL := 'IN_PROGRESS'
COMPSEAM := ''
SEAMQUAL := ''
OPCUAVAR(0,100,COMPTYPE, 'COMPTYPE')
OPCUAVAR(0,99, COMPVARI, 'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI, 'COMPSERNO')
OPCUAVAR(0,97, COMPSTEP, 'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL, 'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
OPCUAVAR(0,93, SEAMQUAL, 'COMPSEAMQUAL')
WAITM (300)
ENDP

```

Этот блок определяет процедуру, передающую на сайт C-Gate информацию о том, что компонент завершен.

```

! КОНЕЦ ЗАПИСИ КОМПОНЕНТА
PUBLIC PROC COMPEND
COMPSEAM := ''
COMPSTAT := 'END'
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
WAITM(1000)

COMPTYPE := ''
COMPVARI := ''
COMPSERI := ''
COMPSTEP := ''
COMPSTAT := ''

```

```

COMPQUAL := ''
COMPSEAM := ''
SEAMQUAL := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99,COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98,COMPSEMI,'COMPSEMI')
OPCUAVAR(0,97,COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96,COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,93,SEAMQUAL,'COMPSEAMQUAL')
ENDP

```

Этот блок определяет функцию, передающую на сайт C-Gate информацию о том, что начинается новый шов. Перед вызовом процедуры имя шва должно быть записано в переменную `COMPSEAM`.

```

! НАЧАЛО ШВА
PUBLIC PROC SEAMSTART
SEAMQUAL := ''
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,93,SEAMQUAL,'COMPSEAMQUAL')
WAITM(100)
ENDP

```

Этот блок определяет функцию, передающую на C-Gate информацию о том, что текущий шов завершен.

```

! КОНЕЦ ШВА
PUBLIC PROC SEAMEND
COMPSEAM := ''
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

```

Этот блок определяет функцию, которая запрашивает на PHG, в порядке ли компонент или нет. Проверка основывается на визуальном осмотре работником и блокирует дальнейший поток программы до тех пор, пока не будет сделана соответствующая запись. Вы можете адаптировать формулировку `ЯВЛЯЕТСЯ ЛИ КОМПОНЕНТ ОК=1 ИЛИ NOT ОК=0?` к своим требованиям.

```

! КАЧЕСТВЕННАЯ ЗАПИСЬ КОМПОНЕНТОВ
PUBLIC PROC COMPQMAN
WREAD ('ЯВЛЯЕТСЯ ЛИ КОМПОНЕНТ ОК=1 ИЛИ NOT ОК=0?', SEL )
IF SEL=1 THEN COMPQUAL := 'ОК' ELSE COMPQUAL := 'NOT_OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP

```

Этот блок определяет функцию, которая автоматически (без запроса) устанавливает качество компонента на ОК. Эта функция может быть использована, если вы не хотите динамически записывать качество общей эффективности оборудования (OEE), но при этом хотите, чтобы вся метрика рассчитывалась корректно.

```
! КАЧЕСТВЕННАЯ ЗАПИСЬ КОМПОНЕНТОВ
PUBLIC PROC COMPQAUT
COMPQUAL := 'OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP
```

Этот блок определяет функцию, которая проверяет, установлена ли переменная 'SEAMQUAL'. Если она не установлена, то качество шва устанавливается на "не ОК". Если установлена, то качество шва устанавливается на соответствующее значение. Переменная 'SEAMQUAL' должна быть записана до вызова этой процедуры.

```
! ОК NOT_OK КАЧЕСТВО ТЕКУЩЕГО ШВА. ДОЛЖНО БЫТЬ УСТАНОВЛЕНО ДЛЯ ХОРОШЕГО КАЧЕСТВА ШВА
И ДЛЯ ПЛОХОГО КАЧЕСТВА.
PUBLIC PROC SEAMQAUT
STRHND(2,SEAMQUAL,SEAMQUAL,RESULT)
IF RESULT=0 THEN BEGIN
WRITE ('ПЕРЕМЕННАЯ >SEAMQUAL< ЕЩЕ НЕ ОПИСАНА!')
WAITM (500)
PAUSE
SEAMQUAL := 'NOT_OK'
END
OPCUAVAR(0,93,SEAMQUAL,'COMPSEAMQUAL')
WAITM (300)
ENDP
```

2.1.2 Переменные

В программе можно использовать следующие переменные.

- **COMPTYPE** : Тип компонента.
- **COMPVARI** : Вариант компонента.
- **COMPSEAM** : Серийный номер компонента. Если значение не задано, то C-Gate автоматически создает серийный номер на основе метки времени.
- **COMPSTEP** : Название этапа производства. Значение имеет значение, если компонент обрабатывается на нескольких станциях.
- **COMPSEAM** : Название шва. Шов может состоять из нескольких сегментов полотна.
- **SEAMQUAL** : Качество шва. Может использоваться для учета результатов внешних измерений.

2.1.3 Простая компонентная программа

Пожалуйста, импортируйте переменные и процедуры из магистерской программы.

RESTART

```
! ИМПОРТ ПЕРЕМЕННЫХ И ПРОЦЕДУР
```

```
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEMI,COMPSTEP,COMPSEAM,SEAMQUAL FROM  
MASTER
```

```
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTART,SEAMEND,SEAMQAUT FROM  
MASTER
```

MAIN

В последовательности программ задайте как минимум обязательные поля и, при необходимости, необязательные.

COMPSTART Как только все поля будут установлены, вызовите процедуру .C-Gate Если вы приобрели модуль .QM, вы также можете определить швы и более детально структурировать запись компонента. **SEAMSTART SEAMEND** Для этого используйте процедуры и . **SEAMQAUT** При необходимости качество шва можно задать вручную с помощью процедуры . Завершить шов можно с помощью процедуры **SEAMEND**. Carola **COMPQAUT COMPQMAN** Как только компонент будет готов и вы захотите задать качество компонента из программы, вызовите либо или для передачи данных о качестве. C-Gate Если вы приобрели модуль .QM, качество компонента определяется автоматически с помощью мониторинга QIROX SD. **COMPQAUT COMPQMAN** Вызов процедуры или не влияет на протокол компонента.

Для завершения работы компонента необходимо вызвать процедуру **COMPEND** . Порядок вызовов важен.

Обратите внимание, что между вызовом **COMPEND** и перезапуском компонента с помощью **COMPSTART** должно пройти не более одной секунды, иначе конечный сигнал может быть потерян из-за синхронизации данных в контроллере. Планируется, что это ограничение будет снято в одной из ближайших версий программного обеспечения для контроллера робота.

```
! УСТАНОВКА ИНФОРМАЦИИ О КОМПОНЕНТЕ
```

```
! ОБЯЗАТЕЛЬНОЕ ПОЛЕ
```

```
COMPTYPE := 'ШКАФ УПРАВЛЕНИЯ'
```

```
! ДОПОЛНИТЕЛЬНО
```

```
! COMPVARI := 'XL'
```

```
! ДОПОЛНИТЕЛЬНО
```

```
! COMPSEMI := '1234567890'
```

```
! ДОПОЛНИТЕЛЬНО
```

```
! COMPSTEP := 'ПОЛ'
```

```
! НАЧАЛО КОМПОНЕНТА
```

```
CALL COMPSTART
```

```
! *****
```

```

! КОМПОНЕНТНАЯ ПРОГРАММА
! *****

COMPSEAM := 'ФИЛЕЙНЫЙ ШОВ 1'
CALL SEAMSTART

! *****
! ПРОГРАММА ФИЛЕЙНЫЙ ШОВ 1. НАПРИМЕР, РАЗЛИЧНЫЕ ДВИЖЕНИЯ GC.
! *****

! ДОПОЛНИТЕЛЬНО
! IF CONDITION THEN SEAMQUAL := 'ОК' ELSE SEAMQUAL := 'NOT_ОК'
! CALL SEAMQAUT

COMPSEAM := 'ФИЛЕЙНЫЙ СВАРНОЙ ШОВ 2'
CALL SEAMSTART

! *****
! ПРОГРАММА ФИЛЕЙНЫЙ СВАРНОЙ ШОВ 2. НАПРИМЕР, РАЗЛИЧНЫЕ ДВИЖЕНИЯ GC.
!
! ДОПОЛНИТЕЛЬНО
! SEAMQUAL := 'NOT_ОК'
! CALL SEAMQAUT
! *****

CALL SEAMEND

! УСТАНОВИТЕ КАЧЕСТВО КОМПОНЕНТА В ПРОГРАММЕ CAROLA
! ЗВОНИТЕ ДО COMPEND
CALL COMPQMAN
! ИЛИ
! CALL COMPQAUT

! КОНЕЦ КОМПОНЕНТА
! ЗВОНИТЕ ПО АДРЕСУ COMPQMAN ИЛИ COMPQAUT
CALL COMPEND

! ОКОНЧАНИЕ ПРОГРАММЫ КОМПОНЕНТА
END

```

2.1.4 Комплексная компонентная программа

Следующая программа Carola создает деталь с двумя производственными этапами. Сварка может производиться как последовательно на одном роботе, так и параллельно на нескольких. Необходимым условием для этого является объединение этапов в протокол:

- Тип, вариант и серийный номер должны быть установлены одинаково для всех роботов.
- В соответствующем шаблоне компонента должно быть задано правильное минимальное количество шагов производства. В данном случае свариваются две производственные ступени, поэтому их количество должно быть установлено на уровне не менее двух.

RESTART

EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEAM,SEAMQUAL FROM MASTER

EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTART,SEAMEND,SEAMQAUT FROM MASTER

MAIN

! *****

! ПРОИЗВОДСТВЕННЫЙ ЭТАП 1

! *****

COMPTYPE := 'КОТЕЛ'

COMPVARI := '80 КВТ'

COMPSEAM := '0042012342'

COMPSTEP := 'ПОЛ'

CALL COMPSTART

COMPSEAM := 'ФИЛЕЙНЫЙ ШОВ 1'

CALL SEAMSTART

! ПРОГРАММА ФИЛЕЙНЫЙ ШОВ 1. НАПРИМЕР, РАЗЛИЧНЫЕ ДВИЖЕНИЯ GC.

COMPSEAM := 'ФИЛЕЙНЫЙ СВАРНОЙ ШОВ 2'

CALL SEAMSTART

! ПРОГРАММА ФИЛЕЙНЫЙ СВАРНОЙ ШОВ 2. НАПРИМЕР, РАЗЛИЧНЫЕ ДВИЖЕНИЯ GC.

CALL SEAMEND

CALL COMPQAUT

CALL COMPEND

! *****

! ПРОИЗВОДСТВЕННЫЙ ЭТАП 2

! *****

COMPTYPE := 'КОТЕЛ'

COMPVARI := '80 КВТ'

COMPSEAM := '0042012342'

COMPSTEP := 'ОБЛОЖКА'

CALL COMPSTART

```

COMPSEAM := 'ФИЛЕЙНЫЙ СВАРНОЙ ШОВ 3'
CALL SEAMSTART
! ПРОГРАММА ФИЛЕЙНЫЙ СВАРНОЙ ШОВ 3. НАПРИМЕР, РАЗЛИЧНЫЕ ДВИЖЕНИЯ GC.

COMPSEAM := 'ФИЛЕЙНЫЙ СВАРНОЙ ШОВ 4'
CALL SEAMSTART
! ПРОГРАММА ФИЛЕЙНЫЙ СВАРНОЙ ШОВ 4. НАПРИМЕР, РАЗЛИЧНЫЕ ДВИЖЕНИЯ GC.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

END

```

3 Примечания

Если вы являетесь или являлись заказчиком PDM или UMS и в конфигурации контроллера активирована одна из двух опций, то необходимо срочно обратить внимание на следующее замечание! Если активирована одна из двух опций, но контроллер робота не подключен к соответствующему приложению ПК, то в потоке данных будут наблюдаться разрывы, которые могут достигать 5 секунд и существенно фальсифицировать протокол компонента. Пожалуйста, либо подключите контроллер робота к приложению ПК, либо удалите соответствующую опцию из конфигурации.

Если вы работаете с несколькими роботами одновременно на одном компоненте и хотите, чтобы все данные были объединены в один протокол компонента в рамках модуля C-Gate.QM, обратитесь в службу CLOOS за дополнительной информацией. В противном случае, пожалуйста, оставьте поле `COMPSEMI` пустым, так как это может привести к нежелательным побочным эффектам при расчете общей эффективности оборудования (OEE).

4 Переменные

Вы имеете возможность присоединить переменные к протоколу компонента из программы Carola. Для этого используется команда `OPCUAVAR` (см. пример). Если имя (последний параметр) имеет префикс `'COMPVAR_'`, то переменная присоединяется к компоненту. Если имя имеет префикс `'STEPVAR_'`, то переменная привязывается к шагу компонента. Если имя имеет префикс `'SEAMVAR_'`, то переменная привязывается к текущему шву. Для каждого выбранного имени может быть сохранено только одно значение. В журнале сохраняется последнее значение, которое было задано для соответствующего имени. Префикс имени не сохраняется. Значение должно быть сброшено перед завершением шва, этапа или компонента. Если этого не сделать, то значение также будет добавлено к следующему элементу. Сохранять можно целые числа, числа через запятую и тексты.

```

CALL SEAMSTART
OPCUAVAR(0,1, 'EXAMPLE-VALUE', 'COMPVAR_SEAM_MEASURE_A')

OPCUAVAR(0,1, '', '')

```

5 Условия

Компонент состоит из нескольких вложенных элементов и может содержать данные от нескольких роботов. Данные роботов записываются в отдельные производственные шаги. Запись производственного шага начинается с процедуры 'COMPSTART' и заканчивается процедурой 'COMPEND'.

Производственный этап состоит из швов. Шов может иметь имя и определяется в программе компонента. Он сохраняется только при записи производственного шага. Он может использоваться для объединения нескольких логически связанных сегментов. Шов начинается и заканчивается с помощью процедур 'SEAMSTRT' и 'SEAMEND'.

Шов состоит из сегментов. Сегмент записывается только тогда, когда робот фиксирует шов и выполняет движение GC или зажигает дугу.

Сегмент состоит из списков. Cada Каждый список сварки, активный в программе во время записи сегмента, также записывается. Это относится только к спискам на основе QTI. Данные сварки записываются и отображаются структурированными по группам или каналам.